
option nomprint ;

libname out "&out_path" ;

%macro compute_consumption ;

%do i = 1 %to %sysfunc(countw(&list_country)) ;
    
    %let country = %scan(&list_country,&i) ;

    %include "&script_path/compute_consumption_&country..sas" ;
    
    proc sql ;
    create table list_work as
    select memname from dictionary.tables where libname = "OUT" ;
    quit ;

    %let c_exist = NO ;

    data list_work ;
    set list_work ;
    if upcase(memname) = "C1" then call symput("c_exist","YES") ;
    run ;

    %if &c_exist = NO %then %do ;

        data out.c1 out.c2 out.c3 out.c4 out.c5 ;
        set h_&country ;
        if im0100 = 1 then output out.c1 ;
        if im0100 = 2 then output out.c2 ;
        if im0100 = 3 then output out.c3 ;
        if im0100 = 4 then output out.c4 ;
        if im0100 = 5 then output out.c5 ;
        keep sa0100 sa0010 im0100 di3001 ;
        run ;
    %end ;
    %else %do ;
        
        proc sort data = h_&country ;
        by sa0100 sa0010 im0100 ;
        run ;

        %do im = 1 %to 5 ;
            proc sort data = out.c&im ;
            by sa0100 sa0010 ;
            run ;

            data out.c&im ;
            update out.c&im h_&country (where = (im0100 = &im) keep 
= sa0100 sa0010 im0100 di3001) ;
            by sa0100 sa0010 ;
            run ;
        %end ;

    %end ;
%end ;

%mend ;



%macro compute_covariates ;
option nonotes nosource ;
/*proc sql noprint ;
select distinct sa0100 into: list_cntr separated by " " from hfcs.h1 
order by sa0100 ;
run ;*/

%let list_var = HW0010 HI0100 HI0200 HB2300 HB0300 HNB0920 HNI0100 
HNI0210 HNI0300 RA0100 RA0200 RA0010 PG0110 PG0210 PG0310 PG0410 
PG0510 PE0100a PA0200 RA0300 DI2000 DH0001 DA3001 DN3001 DHIDH1 ;

proc sql ;
%do n = 1 %to %sysfunc(countw(&list_var)) ;
    create table list&n as
    select * from dictionary.columns where libname = "HFCS" and 
upcase(name) = "%scan(&list_var,&n)" ;
%end ;
quit ;

data list ;
set 
%do n = 1 %to %sysfunc(countw(&list_var)) ;
    list&n 
%end ;
;
run ;

proc datasets library = work nolist ;
delete
%do n = 1 %to %sysfunc(countw(&list_var)) ;
    list&n 
%end ;
;
quit ;

proc sort data = list (keep = memname) out = tab nodupkey ;
by memname ;
run ;

data tab ;
set tab ;
call symput(compress("tab"!!_n_), compress(memname)) ;
call symput("nbtab",_n_) ;
run ;

%do i = 1 %to %sysfunc(countw(&list_country)) ;
    
    %let country = %scan(&list_country,&i) ;
    %put Computing covariates for %upcase(&country) ; 

    %do k = 1 %to &nbtab ;

        data tab&k ;
        set list ;
        where memname = "&&tab&k" ;



        run ;

        proc sql noprint ;
        select name into: var separated by " " from tab&k ;
        quit ;

        data &&tab&k ;
        set hfcs.&&tab&k ;
        keep id sa0010 sa0100 im0100 &var ;
        where sa0100 = "%upcase(&country)" ;
        run ;

    %end ;

    data d ;
    set d1 d2 d3 d4 d5 ;
    run ;

    data h ;
    set h1 h2 h3 h4 h5 ;
    run ;

    data hn ;
    set hn1 hn2 hn3 hn4 hn5 ;
    run ;

    data p ;
    set p1 p2 p3 p4 p5 ;
    run ;

    proc sort data = d ;
    by im0100 sa0100 sa0010 ;
    proc sort data = h ;
    by im0100 sa0100 sa0010 ;
    proc sort data = hn ;
    by im0100 sa0100 sa0010 ;
    run ;

    data h ;
    merge h d hn ;
    by im0100 sa0100 sa0010 ;
    run ;

    /* counting number of children */

    proc sql ;
    create table number_children as
    select distinct sa0010, sa0100, im0100, sum(RA0300 <= 18) as 
number_children from p group by sa0010, sa0100, im0100 ;
    quit ;

    /* counting number of consumption units */

    proc sql ;
    create table number_cu as
    select distinct sa0010, sa0100, im0100, 0.5+sum((RA0300 < 14)*
0.3+(RA0300 ge 14)*0.5) as nbcu from p group by sa0010, sa0100, 
im0100 ;



    quit ;

    proc sql ;
    create table refp as
    select p.* from p, d where p.sa0100 = d.sa0100 and p.sa0010 = 
d.sa0010 and p.ra0010 = d.dhidh1 and p.im0100 = d.im0100 ;
    quit ;

    proc sort data = refp ;
    by im0100 sa0100 sa0010 ;
    proc sort data = number_children ;
    by im0100 sa0100 sa0010 ;
    proc sort data = number_cu ;
    by im0100 sa0100 sa0010 ;
    run ;

    data h ;
    merge h (in=a) refp number_children number_cu ;
    by im0100 sa0100 sa0010 ;
    if a ;
    run ;

    %put Computation done for %upcase(&country) ;

    data h_&country ;
    set h ;
    run ;

%end ;

option notes source ;
%mend ;


